


COMPILING AND RUNNING THE PROGRAMS

DSP Programming – Demonstration Program 3

```
' TestCLI - demonstrate processing audio files from command-line
application
'
' Select a Console application.
'
' Use the "My Project" dialog to configure the program:
' - under the Compile tab, press "Advanced Compilation Options"
'   and change the Target CPU to "x86"
' - under the References tab, click the "Add" button then browse for
'   and select the BasicDSP.dll library.
'
' Run from Command Prompt or CYGWIN:
' TestCLI 1000 inp.wav
' TestCLI 1000 inp.wav out.wav
'
Imports BasicDSP
Module TestCLI
    Sub Main(ByVal args() As String)
        ' usage
        If (args.Count < 2) Then
            Console.Error.WriteLine("usage: TestCLI lpfiltfreq inp.wav
(out.wav)")
            Environment.Exit(1)
        End If
        ' get low-pass filter frequency
        Dim lofreq As Double
        If (Not Double.TryParse(args(0), lofreq)) Then
            Console.Error.WriteLine("unable to parse filter frequency " &
args(0))
            Environment.Exit(1)
        End If
        ' get input wave file
        Dim wv As New Signal
        If (Not wv.LoadWaveFile(args(1))) Then
            Console.Error.WriteLine("unable to read WAV file " & args(1))
            Environment.Exit(1)
        End If
        ' report
        Console.WriteLine("Read " & wv.Count & " samples @ " & wv.Rate & "
from " & args(1))
        ' do filtering
        Dim filt As LTISystemChain =
BasicDSP.Filter.ButterworthLowPass(lofreq / wv.Rate, 4)
        For i As Integer = 0 To wv.Count - 1
            wv(i) = filt(wv(i))
        Next
        ' either replay or save
        If (args.Count < 3) Then
            wv.Replay()
        ElseIf (wv.SaveWaveFile(args(2))) Then
            Console.WriteLine("Wrote " & wv.Count & " samples @ " & wv.Rate
& " to " & args(2))
        Else
            Console.Error.WriteLine("unable to save WAV file " & args(2))
        End If
    End Sub
End Module
```

```
Environment.Exit(1)
End If
Environment.Exit(0)
End Sub
End Module
```



The screenshot shows a Windows command prompt window with the title bar "/cygdrive/c/src/vbasic/basicdsp/testcli/bin/Release". The window contains the following text:

```
Mark@pc128-61 /cygdrive/c/src/vbasic/basicdsp/testcli/bin/Release
$ ./testcli 1000 c:/sfs/demo/six16000.wav
Read 42900 samples @ 16000 from c:/sfs/demo/six16000.wav

Mark@pc128-61 /cygdrive/c/src/vbasic/basicdsp/testcli/bin/Release
$ ./testcli 1000 c:/sfs/demo/six16000.wav out.wav
Read 42900 samples @ 16000 from c:/sfs/demo/six16000.wav
Wrote 42900 samples @ 16000 to out.wav

Mark@pc128-61 /cygdrive/c/src/vbasic/basicdsp/testcli/bin/Release
$
```